
Abstract


The goal of my project is to build a robot that changes its shape when the lights are turned on or off.  The robot must also avoid obstacles while moving around a room in either mode.  By using motors, sensors and a microcontroller, the robot will be able to move around and make decisions based on the surrounding environment.

Executive Summary


Belle is alive but I need to put her body cover on.  Though Belle does have a minimal obstacle avoidance routine, she still collides with obstacles due to a dead spot in between her left and right infrared detectors.  Due to the way I had to drive the electronic speed control, Belle is also moving around but her motor stutters as she moves.  Her motion detector is very good, as she can sense approaching people and even approaching cats.    

Introduction


The desire to build this robot stems from childhood.  When I was younger I played with Gobots and Transformers, and I always wanted them to change shape by themselves.  Therefore, the purpose of this project was to build a robot that will change its shape.  Since the first transformer I owned was a car that turned into a standing figure – my first autonomous robot will do the same.  Further, my first automobile was named Belle so my first autonomous device will also be named Belle. In this paper I will explain the requirements for my robot and how they will be met.


Integrated System

To see the block diagram for the process of transformation and the layout of the platform -- please see the figure marked integrated system. Based on the diagram, the platform satisfies the shape changing criteria of the robot.  

Belle’s brain -- a Motorola 68HC11 connected to the TJ-PRO board, relies on the following software structure to make decisions and perform actions.


a.) A routine to sample all detectors

b.) A routine to make decisions based on the sampled IR values 

c.) A routine to make decisions based on the sampled detector values

d.) Five output compare routines to generate pulse widths

Belle does not have an operating system.  She operates by performing each routine one at a time, and then repeating the loop.  Please see appendix for code.

Mobile Platform


Since the goal was to build a robot the changed its shape from a car to a standing figure I needed a platform that would support this behavior.  Further, the robot had to be able to move around in either mode.  I got help from Randy Wells, a mechanical engineering student, to solve the transformation problem.  


To transform I used a 4-bar linkage system to make Belle stand up.  The original linkage system allowed for Belle’s trunk to fold down during transformation.  Unfortunately, I had to minimize the original linkage system because I could not make Belle short enough.  After I modeled the linkages using MATLAB and cardboard I set out to make the actual platform pieces.  By drawing the platform parts with Auto-Cad and cutting the parts out with the T-Tec machine, the platform was modeled and cut into shape.  For Belle’s drive train I chose the gearbox and universal joints of a radio controlled car.  I purchased these parts from Schumacher.  

The most embarrassing lesson I learned with the platform was that I needed a second set of wheels to keep the robot from tipping backward when it changed shape – what a terrible oversight.  

Actuation


A servo is used to move the legs up or down during the transformation process.  Two servos, one for each arm, are used to move the arms during transformation.  A single servo is used to steer the robot.  A radio controlled car motor is used to drive the robot forward or backward. 


Using the 68HC11’s output compare system I operated all of the servos with 800Hz PWM signals.  By varying the pulse width I relied on the electronics of the servos to move the servomotors to the correct position.  As for the drive motor I used an electronic speed control to move the drive motor.  By varying the pulse width it received, I could make the motor move forward or backward.


The most difficultly had in this phase, was developing a good method for controlling the electronic speed control.  When I adjusted its internal pots for very slow forward and reverse motion it would gradually speed up.  After a few minutes, Belle would be moving too quickly for the IR detectors to react.  Then she would crash into objects.  To solve this problem I decided to send the speed control a 15ms burst of pulses to drive the motor and then I would send the speed control a pulse width that would return it to neutral.  This solution worked but Belle stutters as she moves.   

Sensors


In order to help the robot change shape when the lighting changes, a photo-resistor is used.  To avoid obstacles I am using infrared detectors.  To signal the robot when someone is approaching, I used a motion detector.


a.) A photo-resistor is used to monitor the intensity of the lighting in the room.  I used a voltage divider circuit similar to the one we used in class.  When the lights are on, the A/D value is 82.  When the lights are off and the CDS cell is covered the A/D value is 207.  If the A/D value to detect lighting rises above 150, the robot will be in the standing mode.  The robot will be in the car mode if the A/D value is below 150.  I verified the photo-resistor operation by running the sensor test program available on ICC. 

b.) I can approximate the distance between the sensor and the wall by using hacked infrared detectors (Sharp #GP1U5).  The outputs of the IR detectors give analog values relative to the amount of IR that is received.  Then the A/D samples the detected values and gives the microprocessor a number it can use.  Since the sensitivity of an IR detector is limited by the surrounding environment and the intensity of the transmitted IR signal, I verified the detectors by running the sensor test program available on ICC.  When using the 40kHz output signal from the TJ-Pro board and a hacked Sharp IR detector at different ranges, I got the following results:

c.) A motion detector is used to monitor people approaching the robot.  The motion detector senses the movement
 of heat across its pyro-detector.  When no movement is detected, it gives a 0 volt output, when motion is detected, the output rises to 5 volts.  Since the output of the motion detector is connected to the input capture system, the microprocessor only responds to the rising edge of the motion detectors output. 

I hacked a Radio Shack motion detector (#49-425), to detect motion in front of the robot.  I hacked the motion detector to save the time, rather than build a supporting circuit for the pyro-detector.  To perform the hack, I used an oscilloscope to find the output when the motion sensor is triggered.  I found two pins that respond to changes in motion.  Pin 17 on its micro-controller, rests at 0V and then rises to 5V when motion is detected.  Pin 16 works in the opposite manner.  After connecting the sensor the output of pin #17, and connecting the motion detector’s ground to my board’s ground -- I verified the hack by running the sensor test program available on ICC.  While performing the test with the micro controller’s A/D system, I got the following range of values 

	
	Pin #17
	Pin #16

	No motion detected
	255 
	0-1

	Motion detected
	0-1 
	255


According to Radio Shack, the range of the detector is 30 feet and the angle of detection is 60 degrees.  After testing, I found the threshold of the detector, was 15 feet straight ahead or 10 feet and 20 degrees off center.

The drawback to using this motion detector is that it costs $24.00.  The advantages to this motion detector include, very simple hack procedure, and the detector runs off a 9-volt battery.

Behaviors

Belle’s has a simple set of behaviors.  She will change her shape based on the lighting conditions or she will change her shape when humans approach her.  She can also move around and avoid obstacles in car mode or in the standing mode.    

The most difficult part was getting Belle’s obstacle avoidance to work correctly.  Once I shielded the IR emitters, the obstacle avoidance worked better.  The most embarrassing lesson I learned was that the motion detector will detect movement if its position changes quickly.  To correct this mistake I mounted the motion detector to the hood of the car because it does not move during transformation.   

Experimental Layout and Results

For my minimal obstacle avoidance program I chose to subtract the sampled values of the detectors from one another.  If one side had a considerably larger reading than the other, then my obstacle avoidance routine would turn the wheels.  Since both detectors respond differently to the same stimulus, I made sure that when the two values were subtracted there was a large enough difference between the two values to avoid false detections.   

Since the motion detector worked like a switch, I chose to use the input capture system to monitor detections.   

	
	Pin #17
	Pin #16

	No motion detected
	255 
	0-1

	Motion detected
	0-1 
	255


Conclusion


After demonstrating Belle in lab she does meet all of my initial specifications.  However, her obstacle avoidance needs to be improved, and her speed control needs to be upgraded in order for her to move around without getting confused.  With more time and sensors, I will be able to develop a robust obstacle avoidance algorithm.  I could have allowed myself more time if I finalized my platform sooner.  During the semester, I made platform changes all the way into November, which gave me very little time to write software.  I also satisfied a personal goal, which was to use assembly language to program Belle.  To accomplish my goal, I had to use a combination of my EVBU board from EEL4744, and a trial and error method to write Belle’s code.  I could not get PC Bug11 to work with the 68HC11A1. 


In the immediate future I would like to mount the lexan car body to Belle’s chassis.  Then I would like to shorten Belle’s wheelbase as a standing figure.  Finally I would like to build a next generation robot that can drive around as a car and then stand up and walk.

Documentation

Fred Martin, The 6.270 Robot Builder’s Guide 2nd edition, MIT Media Lab, Cambridge, MA, 1992.

Joseph Jones, Bruce Seiger and Anita Flynn, Mobile Robots: Inspiration to implementation 2nd edition, A.K. Peters Publishers, Nattick, MA, 1998

Keith L. Doty.
TJ-PRO Assembly Manual; Mekatronix 1999.

IMDL class: Instruction from Dr. Arroyo, Dr. Schwartz, Rand Chandler, and Scott Nortman

IMDL Web Page: http://mil.ufl.edu/imdl/handouts.

Appendices

Please see attached code.

* Title         : Belle robot code

* Filename      : BELLE.ASM 

* Programmer    : JEREMY ANDERSON

* Date          : December 12, 2000

* Version       : Version 1.00

* Description   : This program is designed to control Belle -- the shape

*                 changing robot.  This program uses five output copares.

*                 Four to control servos that manipulate Belle's legs, arms,

*                 and steering. One to control Belle's electronic speed

*                 control.  One input capture is used to help Belle

*                 respond to detected motion.  The A/D converter is used to

*                 help Belle's sensors detect the distance between them and

*                 an object.  As well as when the photo-resistor detects a

*                 lighting change.  Seperate subroutines are used to gather

*                 the detetor values, avoid obstacles, change shape, move

*                 forward, move backwrd, waste time (delays), and detect

*                 motion.

************************************************************************

* NOTE:   The number of lines in the comments does not necessarily 

*          correspond to the number of lines in the program code.

************************************************************************

**************************************************************************

* Define the address locations of the various registers and user-defined

* constants used in the program

**************************************************************************

*

BAUD    EQU     $102B  ; Baud-rate control register to set the baud rate 

OC1M    EQU     $100C  ; Baud-rate control register to set the baud rate 

OC1D    EQU     $100D  ; Baud-rate control register to set the baud rate

SCCR1   EQU     $102C  ; Serial Communication Control Register-1

SCCR2   EQU     $102D  ; Serial Communication Control Register-2

SCSR    EQU     $102E  ; Serial Communication Status Register

SCDR    EQU     $102F  ; Serial Communication Data Register

OPTION  EQU     $1039  ; Used to power up the A/D

PACTL   EQU     $1026  ; Pulse accumulator control

TOC1    EQU     $1016  ; OC2 Timer output compare register

TOC2    EQU     $1018  ; OC2 Timer output compare register

TOC3    EQU     $101A  ; OC3 Timer output compare register

TOC4    EQU     $101C  ; OC4 Timer output compare register

TOC5    EQU     $101E  ; OC5 Timer output compare register

TCTL1   EQU     $1020  ; Timer control 1 register

TCTL2   EQU     $1021  ; Timer control 1 register

TFLG1   EQU     $1023  ; Timer interrupt flag register

TMSK1   EQU     $1022  ; Timer interrupt mask register

ADR1    EQU     $1031  ; Address 2 conversion register

ADR2    EQU     $1032  ; Address 3 conversion register

ADR3    EQU     $1033  ; Address 2 conversion register

ADR4    EQU     $1034  ; Address 3 conversion register

ADCTL   EQU     $1030  ; A/D control status register

EOS     EQU     $04    ; Unused

CR      EQU     $0D    ; Unused

LF      EQU     $0A    ; Unused

FLUFF   EQU     100    ; Unused

        ORG     $8000

VOLTAGE RMB     1      ; Unused variable

VOLTAG1 RMB     1      ; Unused variable

RIR     RMB     1      ; The converted input voltage of the right IR detector

LIR     RMB     1      ; The converted input voltage of the left IR detector

BUMP    RMB     1      ; The converted voltage of the bump detector  UNUSED

PHOTO   RMB     1      ; The converted input voltage of the light detector

PRIOR   RMB     1      ; Unused variable

ESC     FDB     2      ; Variable for the high part of the PWM - speed control

Temp1   RMB     1      ; Unused variable

VENDA   RMB     2      ; Unused variable

FENDA   RMB     2      ; Unused variable

LVL     RMB     1      ; Unused variable

PWMP1P  RMB     2      ; Variable for the number of counts to add 

OFFHI   RMB     2      ; Variable for the number of counts to add 

OFFLO   RMB     2      ; Variable for the number of counts to add 

WELP    RMB     1      ; Unused variable

BUFF    RMB     1      ; Unused variable

KOUNT   FDB     2      ; Variable used as a 16 bit counter 

COUNT   RMB     1      ; Variable used as an 8 bit counter

PWMHI   RMB     2      ; Unused variable

PWM     RMB     2      ; Unused variable

L_ARM   FDB     2      ; Variable for the high part of the PWM - left arm 

R_ARM   FDB     2      ; Variable for the high part of the PWM - right arm

STEER   FDB     2      ; Variable for the high part of the PWM - steering

LEGS    FDB     2      ; Variable for the high part of the PWM - legs

CENTR   FDB     2      ; Unused variable

************************************************************************

* Initialize Interrupt Jump Vectors

************************************************************************

        ORG     $FFFE           ; Reset jump vector.  Upon reset 

        FDB     Main            ; go to the start of the program

        ORG     $FFE0           ; TOC5/PA3 interrupt vector

        FDB     LG_ISR          ; to move Belle's legs 

        ORG     $FFE2           ; TOC4/PA4 interrupt vector

        FDB     ST_ISR          ; to control Belle's steering

        ORG     $FFE4           ; TOC3/PA5 interrupt vector

        FDB     RA_ISR          ; to move Belle's right arm

        ORG     $FFE6           ; TOC2/PA6 interrupt vector

        FDB     LA_ISR          ; to move Belle's left arm

        ORG     $FFE8           ; TOC1/PA7 interrupt vector

        FDB     ESCISR          ; to move Belle's speed control 

        ORG     $FFEC           ; IC2 interrupt vector used to

        FDB     IC2_ISR         ; monitor Belle's motion detector. 

******************************************************************************

*                       MAIN PROGRAM

* Description: This program calls the InitSCI routine, prints your text,

*              and your list.  Then, it returns to BUFFALO. 

* Calls: samps, AVOID, Delay, DETECT 

******************************************************************************

*

        ORG     $9000           ; Start of Main program

Main    LDS     #$A000          ; Define a stack 

        LDAA    #%10000000      ; Set up the data direction for 

        STAA    PACTL           ; Port A bit 7 (output)

        LDAA    #%10000000      ; Power up the A/D

        STAA    OPTION

        LDAA    #%00010000      ; Set up the A/D for single

        STAA    ADCTL           ; channel multiscan

        LDD     #3700           ; Set the pulse width so that the 

        STD     TOC1            ; speed control starts in neutral

        std     ESC             ; 

        LDD     #4400           ; Set the pulse width so that the 

        STD     TOC2            ; left arm starts in the down 

        std     L_ARM           ; position for car mode.

        LDD     #2100           ; Set the pulse width so that the 

        STD     TOC3            ; right arm starts in the down 

        std     R_ARM           ; position for car mode.

        LDD     #3250           ; Set the pulse width so that the 

        STD     TOC4            ; steering starts out centered

        std     STEER

        LDD     #4800           ; Set the pulse width so that the 

        STD     TOC5            ; legs start in the down position

        std     LEGS            ; for car mode.

        LDAA    #0              ; Use zero to initially clear the 

        STAA    RIR             ; right infrared sensor variable the

        STAA    LIR             ; left infrared sensor variable and 

        STAA    PHOTO           ; the photo resistor sensor variable

        LDD     #0              ; Use zero to initially clear the 

        STD     KOUNT           ; two bit counting variable

        LDAA    #0              ; Use zero to initially clear the 

        STAA    COUNT           ; one bit counting variable

        LDAA    #%10000000      ; Set bit seven of the OC1 mask register 

        STAA    OC1M            ; to control PA7 

        LDAA    #%00000000      ; Preset the output of OC1 data register

        STAA    OC1D            ; to low

        LDAA    #%10101010      ; Set OC2 OC3 OC4 OC5 to clear the

        STAA    TCTL1           ; output pin

        LDAA    #%00000100      ; Set IC2 to capture on rising edges

        STAA    TCTL2           ; to enter interrupt

        LDAA    #%11111010      ; Clear the OC1 OC2 OC3 OC4 OC5 IC2

        STAA    TFLG1           ; interrupt flags 

        LDAA    #%11111010      ; Enable interrupt for

        STAA    TMSK1           ; OC1 OC2 OC3 OC4 OC5 IC2

        jsr     FWD             ; Call the subroutine to get 

        ldaa    #15             ; Belle moving forward

        staa    PRIOR           ; Currently not used

        CLI                     ; Turn on the interrupt system

HERE    LDAA    #$FF            ; Set all of the bits to turn on

        STAA    $7000           ; the 40kHz LED enitters

        jsr     samps           ; Sample all of the sensors 

        jsr     Delay           ; Run the delay to pad the time sample time

        jsr     samps           ; Sample sensors again to get better values 

        JSR     AVOID           ; Call the obstacle avoidance subroutine

        JSR     Detect          ; Call the motion and light detection routine

        jsr     Delay           ; Run the delay to pad the time sample time

        bra     HERE            ; Restart the loop

********************************************************************************

* This interrupt service routine uses OC2 (PA6), to generate a PWM signal

* to control the servo that moves Belle's left arm.

* The frequency of the PWM signal is 800Hz.

********************************************************************************

LA_ISR  LDAA    TFLG1           ; Check if the OC2 flag is set

        ANDA    #%01000000      ; if not get out

        BEQ     END_WAV         ; 

        LDAA    TCTL1           ; Check bit 6 of TCTL1   

        ANDA    #%01000000      ; if it is not set then it must be the

        BEQ     LOW             ; low time of the pulse so branch to low

        LDAA    TCTL1           ; If here it must be the high time

        ANDA    #%11000000      ; so toggle output high PA6 = 1 

        ORAA    TCTL1           ; Use OR so other routines are not effected

        LDD     TOC2            ; Get the current time of TOC2

        ADDD    L_ARM           ; Add the high time for the pulse

        STD     TOC2            ; Set the new value as the next TOC2 time

        BRA     ROUND           ; Go to the end of the routine

LOW     LDD     TOC2            ; Get the current time of TOC2

        ADDD    #40000-L_ARM    ; calculate the low time for the pulse

        STD     TOC2            ; Set the new value as the next TOC2 time

        LDAA    TCTL1           ; Get the present value of TCTL1

        ANDA    #%10000000      ; so toggle the output low PA6 = 0 

        ORAA    TCTL1           ; Use OR so other routines are not effected

ROUND   LDAA    TCTL1           ; Get the current value of TCTL1

        EORA    #%01000000      ; Toggle or clear TCTL1

        STAA    TCTL1           ; Toggle output of PA6

END_WAV LDAA    #%01000000      ; Clear the OC2 interrupt flag before

        STAA    TFLG1           ; leaving the service routine

        RTI                     ; Return from interrupt

******************************************************************************

* This interrupt service routine uses OC3 (PA5), to generate a PWM signal

* to control the servo that moves Belle's right arm.

* The frequency of the PWM signal is 800Hz.

********************************************************************************

RA_ISR  LDAA    TFLG1           ; Check if the OC3 flag is set

        ANDA    #%00100000      ; if not get out

        BEQ     end_wav         ; 

        LDAA    TCTL1           ; Check bit 4 of TCTL1   

        ANDA    #%00010000      ; if it is not set then it must be the

        BEQ     low             ; low time of the pulse so branch to low

        LDAA    TCTL1           ; If here it must be the high time

        ANDA    #%00110000      ; so toggle output high PA5 = 1 

        orAA    TCTL1           ; Use OR so other routines are not effected

        LDD     TOC3            ; Get the current time of TOC3

        ADDD    R_ARM           ; Add the high time for the pulse

        STD     TOC3            ; Set the new value as the next TOC3 time

        BRA     round           ; Go to the end of the routine

low     LDD     TOC3            ; Get the current time of TOC3

        ADDD    #40000-R_ARM    ; calculate the low time for the pulse

        STD     TOC3            ; Set the new value as the next TOC3 time

        LDAA    TCTL1           ; Get the present value of TCTL1

        ANDA    #%00100000      ; so toggle the output low PA5 = 0 

        orAA    TCTL1           ; Use OR so other routines are not effected

round   LDAA    TCTL1           ; Get the current value of TCTL1

        EORA    #%00010000      ; Toggle or clear TCTL1

        STAA    TCTL1           ; Toggle output of PA5

end_wav LDAA    #%00100000      ; Clear the OC3 flag before leaving

        STAA    TFLG1           ; the interrupt service routine

        RTI                     ; Return from interrupt

*******************************************************************************

* This interrupt service routine uses OC4 (PA4), to generate a PWM signal

* to control the servo that controls Belle's steering.

* The frequency of the PWM signal is 800Hz.

********************************************************************************

ST_ISR  LDAA    TFLG1           ; Check if the OC4 flag is set

        ANDA    #%00010000      ; if not get out

        BEQ     End_wav         ; 

        LDAA    TCTL1           ; Check bit 4 of TCTL1   

        ANDA    #%00000100      ; if it is not set then it must be the

        BEQ     Low             ; low time of the pulse so branch to low

        LDAA    TCTL1           ; If here it must be the high time

        ANDA    #%00001100      ; so toggle output high PA4 = 1 

        orAA    TCTL1           ; Use OR so other routines are not effected

        LDD     TOC4            ; Get the current time of TOC4

        ADDD    STEER           ; Add the high time for the pulse

        STD     TOC4            ; Set the new value as the next TOC4 time

        BRA     Round           ; Go to the end of the routine

Low     LDD     TOC4            ; Get the current time of TOC4

        ADDD    #40000-STEER    ; calculate the low time for the pulse

        STD     TOC4            ; Set the new value as the next TOC4 time

        LDAA    TCTL1           ; Get the present value of TCTL1

        ANDA    #%00001000      ; so toggle the output low PA4 = 0 

        orAA    TCTL1           ; Use OR so other routines are not effected

Round   LDAA    TCTL1           ; Get the current value of TCTL1

        EORA    #%00000100      ; Toggle or clear TCTL1

        STAA    TCTL1           ; Toggle output of PA4

End_wav LDAA    #%00010000      ; Clear the OC4 flag before leaving

        STAA    TFLG1           ; the interrupt service routine

        RTI                     ; Return from interrupt

********************************************************************************

* This interrupt service routine uses OC5 (PA3), to generate a PWM signal

* to control the servo that moves Belle's legs.

* The frequency of the PWM signal is 800Hz.

********************************************************************************

LG_ISR  LDAA    TFLG1           ; Check if the OC5 flag is set

        ANDA    #%00001000      ; if not get out

        BEQ     End_waV         ; 

        LDAA    TCTL1           ; Check bit 0 of TCTL1   

        ANDA    #%00000001      ; if it is not set then it must be the

        BEQ     LoW             ; low time of the pulse so branch to low

        LDAA    TCTL1           ; If here it must be the high time

        ANDA    #%00000011      ; so toggle output high PA3 = 1 

        orAA    TCTL1           ; Use OR so other routines are not effected

        LDD     TOC5            ; Get the current time of TOC5

        ADDD    LEGS            ; Add the high time for the pulse

        STD     TOC5            ; Set the new value as the next TOC5 time

        BRA     RounD           ; Go to the end of the routine

LoW     LDD     TOC5            ; Get the current time of TOC5

        ADDD    #40000-LEGS     ; calculate the low time for the pulse

        STD     TOC5            ; Set the new value as the next TOC5 time

        LDAA    TCTL1           ; Get the present value of TCTL1

        ANDA    #%00000010      ; so toggle the output low PA3 = 0 

        orAA    TCTL1           ; Use OR so other routines are not effected

RounD   LDAA    TCTL1           ; Get the current value of TCTL1

        EORA    #%00000001      ; Toggle or clear TCTL1

        STAA    TCTL1           ; Toggle output of PA3

        LDAA    #%00000010      ; Enable IC2 without 

        ORAA    TMSK1           ; effecting the other interrupts

End_waV LDAA    #%00001000      ; Clear the OC5 flag before leaving

        STAA    TFLG1           ; the interrupt service routine

        RTI                     ; Return from interrupt

*********************************************************************************

* This interrupt service routine uses OC1 (PA7), to generate a PWM signal

* to control the electronic speed control.

* The frequency of the PWM signal is 800Hz.

********************************************************************************

ESCISR  LDAA    TFLG1           ; Check if the OC1 flag is set

        ANDA    #%10000000      ; if not get out

        BEQ     EndwaV          ; 

        LDAA    OC1D            ; Check bit 7 of TCTL1   

        ANDA    #%10000000      ; if it is not set then it must be the

        BEQ     setlo           ; low time of the pulse so branch to low

        LDD     TOC1            ; If here it must be the high time

        ADDD    ESC             ; Add the high time for the pulse

        STD     TOC1            ; Set the new value as the next TOC1 time

        BRA     EndwaV          ; Go to the end of the routine

setlo   ldd     TOC1            ; Get the current time of TOC1

        ADDD    #40000-ESC      ; calculate the low time for the pulse

        STD     TOC1            ; Set the new value as the next TOC1 time

EndwaV  LDAA    OC1D            ; Get the current value of OC1D

        EORA    #%10000000      ; Toggle or clear OC1D

        STAA    OC1D            ; Toggle output of PA7

        LDAA    #%10000000      ; Clear the OC1 flag before leaving

        STAA    TFLG1           ; the interrupt service routine

        RTI                     ; Return from interrupt

********************************************************************************

* This subroutine is called to change Belle into car mode.

* It sets the high times of the pulse widths that control

* the left arm, the right arm, and the legs. 

******************************************************************************

CARMD   PSHA                    ; Save contents of B accumulator 

        PSHB                    ; Save contents of A accumulator 

        PSHX                    ; Save contents of X register 

        LDD     #4600           ; Set the value of the pulse high time  

        std     L_ARM           ; to move the left arm down for car mode

        LDD     #2100           ; Set the value of the pulse high time  

        std     R_ARM           ; to move the right arm down for car mode

        LDD     #4700           ; Set the value of the pulse high time  

        std     LEGS            ; to move the legs down for car mode

        PULX                    ; Restore the contents of the X register

        PULB                    ; Restore the contents of the B accumulator

        PULA                    ; Restore the contents of the A accumulator

        RTS                     ; Return from subroutine

********************************************************************************

* This subroutine is called to change Belle into standing mode.

* It sets the high times of the pulse widths that control

* the left arm, the right arm, and the legs. 

******************************************************************************

TRNFRM  PSHA                    ; Save contents of B accumulator 

        PSHB                    ; Save contents of A accumulator 

        PSHX                    ; Save contents of X register 

        LDD     #1950           ; Set the value of the pulse high time  

        std     L_ARM           ; to move the left arm up for standing mode

        LDD     #5150           ; Set the value of the pulse high time  

        std     R_ARM           ; to move the right arm up for standing mode

        LDD     #2350           ; Set the value of the pulse high time  

        std     LEGS            ; to move the legs up for standing mode

        PULX                    ; Restore the contents of the X register

        PULB                    ; Restore the contents of the B accumulator

        PULA                    ; Restore the contents of the A accumulator

        RTS                     ; Return from subroutine

********************************************************************************

*       This subroutine serves as a delay loop to buffer the

*       time between taking samples.

********************************************************************************

Delay   PSHB                    ; Save contents of B accumulator 

        PSHA                    ; Save contents of A accumulator 

        PSHX                    ; Save contents of X register 

        PSHY                    ; Save contents of Y register 

        LDaa    #$1             ; Put the value $1 in the A-accum

MAS     LDX     #16000          ; Put the value 16000 in the x-reg        

loop    DEX                     ; Decrement X

        BNE     loop            ; Repeat until X is zero

        DEca                    ; Decrement A

        BNE     MAS             ; Repeat until A is zero

        PULY                    ; Restore the Y register

        PULX                    ; Restore the X register

        PULA                    ; Restore the contents of the A accumulator

        PULB                    ; Restore the contents of the B accumulator

        RTS                     ; Return from subroutine

********************************************************************************

* This subroutine is called to take samples from the first four channels

* of the A/D port.  Then the sampled values are stored into the corresponding

* varibles.  The variables modified include, BUMP, LIR, RIR, and PHOTO.

********************************************************************************

samps   pshx                    ; Save contents of X register 

        pshb                    ; Save contents of B accumulator 

        psha                    ; Save contents of A accumulator 

        LDAA    #%00010000      ; Set up the A/D for single

        STAA    ADCTL           ; channel single scan

CHILL   ldaa    ADCTL           ; Check ADCTL register to see if the  

        ANDA    #%10000000      ; conversion complete flag has been set

        BNE     CHILL           ; If CCF has not been set loop until it has

        ldaa    ADR1            ; Get the sampled value of channel 1

        staa    BUMP            ; and store it into the variable BUMP

        ldaa    ADR2            ; Get the sampled value of channel 2

        staa    LIR             ; and store it into the variable RIR

        ldaa    ADR3            ; Get the sampled value of channel 3

        staa    RIR             ; and store it into the variable LIR

        ldaa    ADR4            ; Get the sampled value of channel 4

        staa    PHOTO           ; and store it into the variable PHOTO

        PULA                    ; Restore the contents of the A accumulator

        PULB                    ; Restore the contents of the B accumulator

        PULX                    ; Restore the contents of the X register

        RTS                     ; Return from subroutine

********************************************************************************

* This subroutine is called to help Belle avoid obstacles and move.

* It call the subroutines that set the pulsewidths for steering control,

* forward movement, and backward movement.

* Calls: REV and FWD

*******************************************************************************

AVOID   pshx                    ; Save contents of X register 

        pshb                    ; Save contents of B accumulator 

        psha                    ; Save contents of A accumulator 

        ldaa    LIR             ; Get the value of the left IR detector

        LDAB    RIR             ; Get the value of the right IR detector

        ADDB    #35             ; Add an offset to see if the difference

        SBA                     ; between left and right is significant

        BLT     NXCK            ; If the left detector sees nothing move on

        LDD     #4850           ; Otherwise set the pulse width to 

        STD     STEER           ; move the steering to the right

        JSR     REV             ; Call the subroutine to move Belle backward 

        BRA     AUDI            ; Go to the end of the subroutine

NXCK    LDAA    RIR             ; Get the value of the right IR detector

        LDAB    LIR             ; Get the value of the left IR detector

        ADDB    #35             ; Add an offset to see if the difference

        SBA                     ; between left and right is significant

        BLT     nxck            ; If the left detector sees nothing move on

        LDD     #2000           ; Otherwise set the pulse width to 

        STD     STEER           ; move the steering to the left

        JSR     REV             ; Call the subroutine to move Belle backward 

        BRA     AUDI            ; Go to the end of the subroutine

nxck    LDD     #3250           ; Otherwise set the pulse width to 

        STD     STEER           ; move the steering to the center

        JSR     FWD             ; Call the subroutine to move Belle forward 

AUDI    PULA                    ; Restore the contents of the A accumulator

        PULB                    ; Restore the contents of the B accumulator

        PULX                    ; Restore the contents of the X register

        RTS                     ; Return from subroutine

********************************************************************************

* This subroutine is called to make Belle move backward.  This is done by

* varying the PWM signal controlling the electronic speed control. First the

* PWM is set to keep the speed control moving backward for 30000 counts

* or 15ms. Then the PWM is set so that the ESC returns to nuetral.  This was

* done because the speed control I used would speed up if it was on too long.

*******************************************************************************

REV     PSHA                    ; Save contents of B accumulator

        PSHB                    ; Save contents of B accumulator 

        PSHX                    ; Save contents of X register 

        LDD     #3875           ; Set the pulse width high time so that the

        std     ESC             ; speed control will make Belle move backward

        ldX    #30000           ; Use X as a counter that starts at 30000 and

run     deX                     ; counts down. This is done because the speed

        bne     run             ; control is unreliable when running. 

        LDD     #3700           ; Set the pulse width high time so that the

        std     ESC             ; speed control is toggled to nuetral

        PULX                    ; Restore the contents of the X register

        PULB                    ; Restore the contents of the B accumulator

        PULA                    ; Restore the contents of the A accumulator

        RTS                     ; Return from subroutine

********************************************************************************

* This subroutine is called to make Belle move forward.  This is done by

* varying the PWM signal controlling the electronic speed control. First the

* PWM is set to keep the speed control moving backward for 30000 counts

* or 15ms. Then the PWM is set so that the ESC returns to nuetral.  This was

* done because the speed control I used would speed up if it was on too long.

*******************************************************************************

FWD     PSHA                    ; Save contents of A accumulator 

        PSHB                    ; Save contents of B accumulator 

        PSHX                    ; Save contents of X register 

        LDD     #3515           ; Set the pulse width high time so that the

        std     ESC             ; speed control will make Belle move forward

        ldX    #30000           ; Use X as a counter that starts at 30000 and

RUN     deX                     ; counts down. This is done because the speed

        bne     RUN             ; control is unreliable when running. 

        LDD     #3700           ; Set the pulse width high time so that the

        std     ESC             ; speed control is toggled to nuetral

        PULX                    ; Restore the contents of the X register

        PULB                    ; Restore the contents of the B accumulator

        PULA                    ; Restore the contents of the A accumulator

        RTS                     ; Return from subroutine

****************************************************************************

* This interrupt service routine uses IC2 (PA1), to detect if motion has

* occurred.  This routine captures on the rising edges sent from the

* motion detector.

********************************************************************************

IC2_ISR LDAA    TFLG1           ; Check if the IC1 flag is set

        ANDA    #%00000010      ; if not get out

        BEQ     OUT_PB

        ldd     #1              ; Set the 16 bit accumulator to 1

        std     KOUNT           ; and set the 16 bit counter KOUNT = 1

        ldaa    #18             ; Set the 8 bit accumulator to 18 or less and

        STAA    COUNT           ; set COUNT=18. Above 18--program will hang

        LDAA    #%00000010      ; Clear the OC1 flag before leaving

        STAA    TFLG1           ; the interrupt service routine

        LDAA    #%11111000      ; Disable IC2 without 

        ANDA    TMSK1           ; effecting the other interrupts

OUT_PB  RTI                     ; Return from interrupt

********************************************************************************

* This subroutine is called to decide if motion or a light change has been

* detected.  If motion has been detected then Belle will remain in the

* mode until COUNT (the counter set in IC2) becomes zero.  Then if the

* photo-resistor has detected a light change the Belle will change to the

* corresponding shape.  Near total darkness is chosen for Belle to change

* into standing mode.  This was done to eliminate the impact of differences

* in lighting for different rooms. 

*******************************************************************************

Detect  PSHA                    ; Save contents of A accumulator

        PSHB                    ; Save contents of B accumulator 

        PSHX                    ; Save contents of X register 

        LDAA    COUNT           ; Get the current value of 8 bit counter

        BEQ     LEAF            ; when the counter is clear move on 

        DECA                    ; otherwise decrement the value in the A-acc

        STAA    COUNT           ; and store it as the new value of the counter

        JSR     TRNFRM          ; Call the routine to make Belle stand 

        BRA     LEAVE           ; After shape change leave the subroutine

LEAF    JSR     CARMD           ; Call the routine to make Belle a car 

        LDAA    PHOTO           ; Get the current value of the PHOTO detector

        SUBA    #253            ; Subtract 253 so that complete dark will

        BLT     CNG             ; make Belle think its dark. Otherwise move on

        JSR     TRNFRM          ; Call the routine to make Belle stand

        BRA     LEAVE           ; After shape change leave the subroutine

CNG     JSR     CARMD           ; Call the routine to make Belle a car 

LEAVE   PULX                    ; Restore the contents of the X register

        PULB                    ; Restore the contents of the B accumulator

        PULA                    ; Restore the contents of the A accumulator

        RTS                     ; Return from subroutine

******************************************************************************

***************************** END OF PROGRAM *********************************

******************************************************************************

